
	Reason for change:
	After submission of R5s220390 an alternative implementation was given, seems that is provoking following 2 issues :

1. Function f_StringSplit must take into account also “&” (Please check structure above) if not included it will provoke a mismatch in match(v_ReceivedDialogId, v_ExpectedDialogId))  as “'Require=replaces” will be part of other parameter as has not been splitted. As an example, charstrings after f_StringSplit with alternative implementation:
 [2] = from-tag=lJ7VXZpAOO2fpD&Require=replaces. this will cause a mismatch ending the test case prematurely.


2.  Alternative propopsal with match(v_ReferToUrl.headers, { cr_GenericParam("Require", "replaces"), * })) will provoke a fail as v_ReferToUrl.headers contains a unique charstring with Callid, From tag, To tag and Require=replaces separated by “;” and “&”. All of them contained in following structure:
     
type union GenValue {
    charstring  tokenOrHost,
    charstring  quotedString
  }

  type record GenericParam
  {
    charstring             id,
    GenValue               paramValue optional
  }

In order to check if “Require=replaces” header field parameter presence, types used in matching mechanisms used must be of template charstringList.

	
	

	Summary of change:
	1. Splitted Replaces content of charstring type, taking as separators “;” and “&”.
Changed types used in matching mechanins to template charstringList to check the presence of “Require=replaces” header field parameter

	
	

	Consequences if not approved:
	A conformant UE may fail the test case.

	Function name
	a_IMS_REFER_ReceiveRequest

	Reason for change
	1. Function f_StringSplit must take into account also “&” if not included it will provoke a mismatch in match(v_ReceivedDialogId, v_ExpectedDialogId))  as “'Require=replaces” will be part of other parameter as has not been splitted.
2. match(v_ReferToUrl.headers, { cr_GenericParam("Require", "replaces"), * })) will provoke a fail as v_ReferToUrl.headers contains a unique charstring with Callid, From tag, To tag and Require=replaces separated by “;” and “&”.


	Summary of change
	1. Splitted Replaces content of charstring type, taking as separators “;” and “&”.
2. Changed types used in matching mechanins to template charstringList to check the presence of “Require=replaces” header field parameter

	TTCN module
	IMS_NR5GC\IMS_Procedures_Conferencing.ttcn

	MCC160 Comment
	



Before Change:
	<<SKIPPED CODE>>
  altstep a_IMS_REFER_ReceiveRequest(out IMS_DATA_REQ p_IMS_REFER_REQ,
                                     charstring p_RequestUriString,
                                     charstring p_ReferToUriString,
                                     template (omit) IMS_PTC_DialogIndex_Type p_ActiveSessionToBeReplaced,
                                     boolean p_ReferCreatingDialog := false,
                                     IMS_CallType_Type p_CallType := CONFERENCE_CALL) runs on IMS_PTC
  { /* @sic R5s141343 change 3.1: p_ReferCreatingDialog := false sic@ */
    /* @sic R5-153765: new parameter p_ActiveSessionToBeReplaced sic@
     * @sic R5s150909: p_ActiveSessionToBeReplaced changed to "template (omit) IMS_PTC_DialogIndex_Type" sic@
     * NOTE: ispresent(p_ActiveSessionToBeReplaced) => there is an active session between the UE and a remote user which shall be replaced (e.g. Three way session creation; 15.121a);
     *                                                 (the remote user is referenced by p_ReferToUriString)
     *       p_ActiveSessionToBeReplaced==omit      => there is no active session with the remote user but the REFER is just to invite the remote user */
     /* @sic R5s160045 p_CSeqValue removed sic@
        @sic R5s180548: p_CallType added sic@
     */
    var IMS_DATA_REQ v_IMS_REFER_REQ_ByRef;
    var template (present) SipUrl v_RequestUriRefer := f_SIP_BuildSipUri_RX(p_RequestUriString);
    var boolean v_ReplacesHeaderIsRequired := ispresent(p_ActiveSessionToBeReplaced);
    var boolean v_RequireHeaderIsRequired := v_ReplacesHeaderIsRequired and (p_CallType == NORMAL_CALL);
    var template (present) SipUrl v_ReferToUri := fl_SIP_BuildReferToUri_RX(p_ReferToUriString, v_ReplacesHeaderIsRequired);  /* @sic R5s150382 change 3; R5s160043, R5-160796 MCC160 implementation  sic@ */
    var template (present) MessageHeader v_MessageHeader_Refer := f_IMS_ReferRequest_MessageHeaderRX(v_RequestUriRefer, v_ReferToUri, p_ReferCreatingDialog, -, p_CallType);
    var SipUrl v_ReferToUrl;
    var template (omit) GenericParam v_Replaces;
    var template (value) CallId v_CallId;
    var charstring v_FromTag;
    var charstring v_ToTag;
    var CharStringList_Type v_ReceivedDialogId;
    var template (present) CharStringList_Type v_ExpectedDialogId;

    [] a_IMS_ReceiveRequest(car_IMS_Refer_Request(cr_REFER_Request(v_RequestUriRefer, v_MessageHeader_Refer)), v_IMS_REFER_REQ_ByRef)
      {
        f_IMS_A_2_10_Refer_CommonChecks(v_IMS_REFER_REQ_ByRef, p_ReferCreatingDialog);
        if (not p_ReferCreatingDialog) {    /* @sic R5s160045: CSeq value to be incremented and checked for in-dialog REFER sic@ */
          f_IMS_RequestInDialog_IncrAndCheckCSeqValue(v_IMS_REFER_REQ_ByRef.Request.Refer.msgHeader);
        }
        v_ReferToUrl := v_IMS_REFER_REQ_ByRef.Request.Refer.msgHeader.referTo.nameAddr.addrSpec;
        if (v_ReplacesHeaderIsRequired) {   /* @sic R5-153765, R5s150909 sic@ */
          // check whether header portion of the SIP URI includes Replaces header if an active session exists @sic R5-153765 sic@
          v_Replaces := f_SIP_AmpersandParamList_GetParam(v_ReferToUrl.headers, "Replaces");
          if (ispresent(v_Replaces)) {
            v_CallId   := f_IMS_PTC_ImsInfo_DialogGetCallId(p_ActiveSessionToBeReplaced);
            v_FromTag  := valueof(f_IMS_PTC_ImsInfo_DialogGetRemoteTag(p_ActiveSessionToBeReplaced));   // @sic R5s150909 sic@
            v_ToTag    := valueof(f_IMS_PTC_ImsInfo_DialogGetLocalTag(p_ActiveSessionToBeReplaced));    // @sic R5s150909 sic@
            v_ExpectedDialogId := {
              v_CallId.callid,
              permutation("to-tag=" & v_ToTag, "from-tag=" & v_FromTag, *)
            };
            v_ReceivedDialogId := f_StringSplit(f_URL_Decoding(valueof(v_Replaces.paramValue.tokenOrHost)), {";"}); // see RFC 3891 6.1
            if (not match(v_ReceivedDialogId, v_ExpectedDialogId)) {
              f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid dialog id in Replaces header of Refer-To's SipUri");
            }
          } else {    /* @sic R5s220390 - additional change sic@ */
            f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Refer-To SipUri does not contain 'Replaces' header field");
          }
        }
        if (v_RequireHeaderIsRequired) {   /* @sic R5-R5s220390 sic@ */
          if (not match(v_ReferToUrl.headers, { cr_GenericParam("Require", "replaces"), * })) {
            f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "'Require=replaces' header field not found");
          }
        }
        p_IMS_REFER_REQ := v_IMS_REFER_REQ_ByRef;  // return message by ref
      }
  } <<SKIPPED CODE>>

	




After Change:
	<<SKIPPED CODE>>
  altstep a_IMS_REFER_ReceiveRequest(out IMS_DATA_REQ p_IMS_REFER_REQ,
                                     charstring p_RequestUriString,
                                     charstring p_ReferToUriString,
                                     template (omit) IMS_PTC_DialogIndex_Type p_ActiveSessionToBeReplaced,
                                     boolean p_ReferCreatingDialog := false,
                                     IMS_CallType_Type p_CallType := CONFERENCE_CALL) runs on IMS_PTC
  { /* @sic R5s141343 change 3.1: p_ReferCreatingDialog := false sic@ */
    /* @sic R5-153765: new parameter p_ActiveSessionToBeReplaced sic@
     * @sic R5s150909: p_ActiveSessionToBeReplaced changed to "template (omit) IMS_PTC_DialogIndex_Type" sic@
     * NOTE: ispresent(p_ActiveSessionToBeReplaced) => there is an active session between the UE and a remote user which shall be replaced (e.g. Three way session creation; 15.121a);
     *                                                 (the remote user is referenced by p_ReferToUriString)
     *       p_ActiveSessionToBeReplaced==omit      => there is no active session with the remote user but the REFER is just to invite the remote user */
     /* @sic R5s160045 p_CSeqValue removed sic@
        @sic R5s180548: p_CallType added sic@
     */
    var IMS_DATA_REQ v_IMS_REFER_REQ_ByRef;
    var template (present) SipUrl v_RequestUriRefer := f_SIP_BuildSipUri_RX(p_RequestUriString);
    var boolean v_ReplacesHeaderIsRequired := ispresent(p_ActiveSessionToBeReplaced);
    var boolean v_RequireHeaderIsRequired := v_ReplacesHeaderIsRequired and (p_CallType == NORMAL_CALL);
    var template (present) SipUrl v_ReferToUri := fl_SIP_BuildReferToUri_RX(p_ReferToUriString, v_ReplacesHeaderIsRequired);  /* @sic R5s150382 change 3; R5s160043, R5-160796 MCC160 implementation  sic@ */
    var template (present) MessageHeader v_MessageHeader_Refer := f_IMS_ReferRequest_MessageHeaderRX(v_RequestUriRefer, v_ReferToUri, p_ReferCreatingDialog, -, p_CallType);
    var SipUrl v_ReferToUrl;
    var template (omit) GenericParam v_Replaces;
    var template (value) CallId v_CallId;
    var charstring v_FromTag;
    var charstring v_ToTag;
    var CharStringList_Type v_ReceivedDialogId;
    var template (present) CharStringList_Type v_ExpectedDialogId;
    var template (present) CharStringList_Type v_ExpectedRequire :=  {*, "Require=replaces"}; 

      
    [] a_IMS_ReceiveRequest(car_IMS_Refer_Request(cr_REFER_Request(v_RequestUriRefer, v_MessageHeader_Refer)), v_IMS_REFER_REQ_ByRef)
      {
        f_IMS_A_2_10_Refer_CommonChecks(v_IMS_REFER_REQ_ByRef, p_ReferCreatingDialog);
        if (not p_ReferCreatingDialog) {    /* @sic R5s160045: CSeq value to be incremented and checked for in-dialog REFER sic@ */
          f_IMS_RequestInDialog_IncrAndCheckCSeqValue(v_IMS_REFER_REQ_ByRef.Request.Refer.msgHeader);
        }
        v_ReferToUrl := v_IMS_REFER_REQ_ByRef.Request.Refer.msgHeader.referTo.nameAddr.addrSpec;
        if (v_ReplacesHeaderIsRequired) {   /* @sic R5-153765, R5s150909 sic@ */
          // check whether header portion of the SIP URI includes Replaces header if an active session exists @sic R5-153765 sic@
          v_Replaces := f_SIP_AmpersandParamList_GetParam(v_ReferToUrl.headers, "Replaces");
          if (ispresent(v_Replaces)) {
            v_CallId   := f_IMS_PTC_ImsInfo_DialogGetCallId(p_ActiveSessionToBeReplaced);
            v_FromTag  := valueof(f_IMS_PTC_ImsInfo_DialogGetRemoteTag(p_ActiveSessionToBeReplaced));   // @sic R5s150909 sic@
            v_ToTag    := valueof(f_IMS_PTC_ImsInfo_DialogGetLocalTag(p_ActiveSessionToBeReplaced));    // @sic R5s150909 sic@
            v_ExpectedDialogId := {
              v_CallId.callid,
              permutation("to-tag=" & v_ToTag, "from-tag=" & v_FromTag, *)
            };
            v_ReceivedDialogId := f_StringSplit(f_URL_Decoding(valueof(v_Replaces.paramValue.tokenOrHost)), {";","&"}); // see RFC 3891 6.1 
            if (not match(v_ReceivedDialogId, v_ExpectedDialogId)) {
              f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid dialog id in Replaces header of Refer-To's SipUri");
            }
          } else {    /* @sic R5s220390 - additional change sic@ */
            f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "Refer-To SipUri does not contain 'Replaces' header field");
          }
        }
        if (v_RequireHeaderIsRequired) {   /* @sic R5-R5s220390 sic@ */
          if (not match(v_ReceivedDialogId, v_ExpectedRequire)) {
            f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "'Require=replaces' header field not found");
          }
        }
        p_IMS_REFER_REQ := v_IMS_REFER_REQ_ByRef;  // return message by ref
      }
  }<<SKIPPED CODE>>





